【Excelでカレンダーを作る】
◆Sheet1とSheet2に下のようなデータを入力します。

〔Sheet1〕

〔Sheet2〕

◆シート名を変更します。

Sheet1
祝日
Sheet2
(1)
>全て半角で入力します
◆(1)シートに各関数を設定します。
	
	関数式
	説　明

	I2
	=TEXT(I1,"aaa")
	曜日を求める

	I3
	=YEAR(I1)
	年を求める

	I4
	=MONTH(I1)
	月を求める

	I5
	=DATE(I3,I4+1,1)-1
	年、月、日を指定して、日付を求める
（+1は、翌月を出すため、
　-1は、前の月の月末の意味）

	I6
	=DAY(I5)
	日を求める

	A1
	=YEAR(I1)&"年"&MONTH(I1)&"月"
	&で数式と文字列をつなぐ（文字列は、""で囲む）


1週目（1日を決めるためのみ）

	
	関数式
	説　明

	A4
	=IF($I$2=A3,1,"")
	同じ曜日ならば、1、違うなら空白（日曜日だけ）

	B4

～G4
	=IF($I$2=B3,1,IF(A4="","",A4+1))
	同じ曜日ならば、1、違うなら空白なら空白、
違うなら前日+1（月～土まで）


2週目以降（月末を決める）

	
	関数式
	説　明

	A7
	=IF(G4="","",IF(G4+1>$I$6,"",G4+1))
	土曜日が空白なら空白、
違うなら前日+1が月末日より大きければ空白、違うなら前日+1（日曜日だけ）

	B7

～G7
	=IF(A7="","",IF(A7+1>$I$6,"",A7+1))
	前日が空白なら空白、
違うなら前日+1が月末より大きければ空白、違うなら前日+1（月～土まで）


A7～G7に設定した数式を

A10～G10

A13～G13

A16～G16

A19～G19

にコピーします。

祝日を設定
下のセル番号に設定します。

A5～G5

A8～G8

A11～G11

A14～G14

A17～G17

A20～G20
=IF(ISERROR(
VLOOKUP(DATE($I$3,$I$4,A4),祝日!$A$2:$B$18,2,0)),"",
VLOOKUP(DATE($I$3,$I$4,A4),祝日!$A$2:$B$18,2,0))

◆各書式設定をします。
A1
：斜体、20pt

曜日
：左右上下中央揃え

曜日の日
：フォントの色　赤

曜日の土
：フォントの色　青

A4～G21
：フォントの色　赤、横位置：左揃え、縦位置：上詰め

列幅
：12

行の高さ
：27

条件付き書式

	
	

	B4～F4

B7～F7

B10～F10

B13～F13

B16～F16

B19～F19
	数式を使用して、書式設定するセルを決定
=IF(B5="",1)

フォントの色：自動

	G4

G7

G10

G13

G16

G19
	数式を使用して、書式設定するセルを決定
=IF(G5="",1)

フォントの色：青


印刷範囲：A1：G21
※最後に、見栄えのよい、罫線を設定すれば1月の出来上がりです。

　シートをコピーして、12か月分を作る。

